code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_40:樹精の森
;-------------------------------------------------
;クエスト選択画面で表示されるクエスト名
@QST_TITLE_QUEST_40
RESULTS '= "樹精の森"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_40
RESULTS '= "周期,探索,"

;クエスト選択画面で呼び出される概要説明
;タグで装飾が可能(@FuncTagSetText_CreateColorBarText 参照)
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_40
RESULTS:0 '= "キャラバン護衛で知り合った商人から名指しの依頼が届いた"
RESULTS:1 '= "辺境地域の片隅にある森で、希少な錬金素材を採取してきてほしいのだという"
RESULTS:2 '= ""
RESULTS:3 '= "訪れるたびに樹々の様相が変わる奇妙な森"
RESULTS:4 '= "樹精の森の通称で知られるその場所について、冒険者ギルドではこんな噂が囁かれている"
RESULTS:5 '= ""
RESULTS:6 '= @"@C:{カラー_警告}@「女泣かせの男でいたいなら、あの森には近付かないほうがいい」@/C@"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_40
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★★ ～ ★★★★★@/C@"
RESULTS:1 '= "報酬：指定品入手数次第"
RESULTS:2 '= "推奨パーティ人数:3人以上　　　推奨平均戦闘値：40以上"
RESULTS:3 '= "備考:定期性クエスト(5期毎)、退却可能"
RESULTS:4 '= @"@C:{カラー_警告}@性技スキル：誘蛾の色香LV2以上の場合は深刻な肉体変化の危険性あり@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「よお、今回も頼むぜ。あそこへ行ってくれそうなのはあんたぐらいしか居なくてな」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「妙な噂のある森だそうだが、なあに、世の中にはよくある大袈裟な迷信さ。だろ？」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_40(QUEST_ID)
#DIM QUEST_ID
;※実行条件に隊商護衛のクリア
SIF !QST_SV_QUEST_FLAG:4
    RETURN 0

;クエストのクールタイム。完了時に加算され、期終了時に減算される
IF QST_COOLTIME:QUEST_ID
    RETURN 0
ELSE
    RETURN 1
ENDIF

;-------------------------------------------------
;マップデータ定義
;[ROOM_ID:]　…　配置するエリアの種類
;[ROOM_PATH:]　…　エリア間の接続情報(※)
;[EVENT_LIST:]　…　エリアに配置するイベントのID(※)
;(※)要素の末尾の','を忘れないように注意
;<マップを編集するには>
;EmueraをDEBUGMODEで起動した場合、[領地の手入れ]画面から編集したエリア情報をテキストファイルにエクスポートできます
;sav/txt00.txtに出力されたマップ情報をコピペして編集すると少し楽になります
;<DEBUG MODE起動方法>
;https://ja.osdn.net/projects/emuera/wiki/debug
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_40(MAP_ID)
#DIM MAP_ID
QST_MAPNAME '= "樹精の森"

;本クエストの狙いとしてはまず女性化クエストを初めとした強制肉体変化イベントの実装
;そして誘蛾の色香を取得していなければ、Ｈの少ない戦闘力上げクエストとしても繰り返し訪れられるクエスト

;報酬は特別ルールで、生還した場合だけお金か名誉に交換できるようにしてありますが、
;報酬額は意図的に少なめにしています（あくまで他クエストの補完、気分転換を目的とするため）
;また、敗北してもどこの部位の処女も失わない、処女性に拘りがある方も安心して負けられるクエストとしています（ボス戦は別）

;北側と南側で遭遇する敵の強さを大きく変えてあります
;南側は、隊商護衛ノーマルはクリアできるが平原の探索はクリアできない程度の低戦闘力帯
;北側は、墳墓の小迷宮ノーマルでは物足りない、気分転換をしたいというときの中戦闘力帯
;それぞれの戦闘力帯の訓練のために使って貰うことを考えています
;ボスは深森の忘都のボスよりは弱い程度のチャレンジボスを想定しています

;ただしただし……現時点でMOBの強さはかなり適当です！
;強すぎ！　弱すぎ！　ついでにテキスト長すぎ！　などなどありましたら意見を頂けますと幸いです（本クエスト制作者：aria）

;本クエストは実験的にランダムマップを採用しています
;北と南でそれぞれ単純２０％、やや単純４０％、やや複雑３０％、複雑１０％の４種類ずつ
;合計パターンは１６種類！　キミだけの樹精の森を探索しよう！

;以下、イベントの説明（イベント番号がいまいちまとまってません……）
;全体的に男性や男の娘が女性に近付いていく性偏差上昇イベントが発生します
;ただしそれを含めた肉体変化系のイベント効果は、誘蛾の芳香レベル２以上でなければ無効、あるいはダメージや別の効果に置き換わります

;発生確率100%のもの
;EV0:女王の閨：ボス戦闘。敗北すると、このクエストで唯一、破瓜する調教が発生します
;EV1-2：穏やかな川：マップ中央の休憩回復
;EV3,5：隘路：採取イベント。成功すると報酬＋、失敗するとダメージか短時間調教イベント
;EV4,6：獣道、隘路、泥地、上り坂、下り坂：雑魚戦闘。負けても処女は奪われません
;EV12：林檎の樹：Ｈ調教を受けずに性偏差を貯めるためのイベント。条件次第でそれとは別に体格減少も
;EV21：洞穴：探索開始すぐに裸にして発情させるための場所
;EV22：岩壁：鉱石入手イベント。５％で金鉱石ひとつ

;確率で発生するもの
;EV7：小屋：30％で発生。マップヒントテキスト的な
;EV8：池：20%で発生。料理判定で失敗すると体格減少、成功すると体格増加も選択できる（長身まで）
;EV9：空き地：20%で発生。特に意味もないフレーバーイベント。そのうち何か追加するかも、しないかも
;EV10：吊り橋：40％で発生。何かありそうで何もない場所。ひとつぐらいはこんな場所もいいかなと
;EV11：妖精庭園：40％で発生。強制膨乳、膨尻イベント。歌唱判定で成功すれば小さくもできる
;EV20：花畑：40％で発生。男の娘ふたなり化、女性化、濡れやすい付与のイベント等。濡れやすいの付与には淫堕の宝玉が必要になります
;EV23：樹々の隙間：約40％で発生。現状入手しにくいけどレアでなくても良さそうなものをちょっとだけ入手できる用イベント

;直接は呼び出されない処理まとめ用のもの
;EV25：精算用イベント
;EV40～46：敗北後や調教関係

SELECTCASE RAND:10
    ;ランダムマップ生成北側
    CASE 0 TO 1
        _DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:201][ROOM_PATH:7,][ROOM_NAME:女王の閨][EVENT_LIST:0,]"
        _DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:201][ROOM_PATH:4,8,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:201][ROOM_PATH:3,][ROOM_NAME:小屋][EVENT_LIST:7,]"
        _DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:201][ROOM_PATH:6,][ROOM_NAME:空き地][EVENT_LIST:9,]"
        _DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:201][ROOM_PATH:5,7,][ROOM_NAME:池][EVENT_LIST:8,]"
        _DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][ROOM_PATH:2,6,8,12,][ROOM_NAME:獣道][EVENT_LIST:6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][ROOM_PATH:3,7,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:0]"
    CASE 2 TO 5
        _DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:201][ROOM_PATH:5,][ROOM_NAME:女王の閨][EVENT_LIST:0,]"
        _DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:201][ROOM_PATH:3,7,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:201][ROOM_PATH:2,8,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:201][ROOM_PATH:0,6,][ROOM_NAME:獣道][EVENT_LIST:6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:201][ROOM_PATH:5,7,][ROOM_NAME:泥地][EVENT_LIST:6,]"
        _DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][ROOM_PATH:2,6,12,][ROOM_NAME:獣道][EVENT_LIST:6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][ROOM_PATH:3,][ROOM_NAME:樹々の隙間][EVENT_LIST:23,]"
        _DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:0]"
    CASE 6 TO 8
        _DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:201][ROOM_PATH:1,5,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:201][ROOM_PATH:0,2,][ROOM_NAME:吊り橋][EVENT_LIST:10,]"
        _DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:201][ROOM_PATH:1,3,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:201][ROOM_PATH:2,][ROOM_NAME:女王の閨][EVENT_LIST:0,]"
        _DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:201][ROOM_PATH:0,6,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:201][ROOM_PATH:5,7,][ROOM_NAME:泥地][EVENT_LIST:6,]"
        _DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][ROOM_PATH:6,8,12,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][ROOM_PATH:7,][ROOM_NAME:妖精庭園][EVENT_LIST:11,]"
        _DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:0]"
    CASEELSE
        _DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:201][ROOM_PATH:1,5,][ROOM_NAME:妖精庭園][EVENT_LIST:11,]"
        _DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:201][ROOM_PATH:0,2,6,][ROOM_NAME:吊り橋][EVENT_LIST:10,]"
        _DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:201][ROOM_PATH:1,3,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:201][ROOM_PATH:2,4,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:201][ROOM_PATH:3,9,][ROOM_NAME:小屋][EVENT_LIST:7,]"
        _DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:201][ROOM_PATH:0,6,][ROOM_NAME:女王の閨][EVENT_LIST:0,]"
        _DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:201][ROOM_PATH:1,5,][ROOM_NAME:泥地][EVENT_LIST:6,]"
        _DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][ROOM_PATH:8,12,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][ROOM_PATH:7,9,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
        _DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:201][ROOM_PATH:4,8,][ROOM_NAME:隘路][EVENT_LIST:5,6,12,]"
ENDSELECT
;中央はいつも共通
_DIC_HO_MAPDATA:MAP_ID:10 '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:11 '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:12 '= "[ROOM_ID:201][ROOM_PATH:7,17,][ROOM_NAME:穏やかな川][EVENT_LIST:1,2,]"
_DIC_HO_MAPDATA:MAP_ID:13 '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:14 '= "[ROOM_ID:0]"
SELECTCASE RAND:10
    ;ランダムマップ生成南側
    CASE 0 TO 1
        _DIC_HO_MAPDATA:MAP_ID:15 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:16 '= "[ROOM_ID:201][ROOM_PATH:17,][ROOM_NAME:樹々の隙間][EVENT_LIST:23,12,]"
        _DIC_HO_MAPDATA:MAP_ID:17 '= "[ROOM_ID:201][ROOM_PATH:12,16,22,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:18 '= "[ROOM_ID:201][ROOM_PATH:19,][ROOM_NAME:洞穴][EVENT_LIST:21,]"
        _DIC_HO_MAPDATA:MAP_ID:19 '= "[ROOM_ID:201][ROOM_PATH:18,24,][ROOM_NAME:岩壁][EVENT_LIST:22,]"
        _DIC_HO_MAPDATA:MAP_ID:20 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:21 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:22 '= "[ROOM_ID:200][ROOM_PATH:17,23,][ROOM_NAME:森の入り口]"
        _DIC_HO_MAPDATA:MAP_ID:23 '= "[ROOM_ID:201][ROOM_PATH:22,24,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:24 '= "[ROOM_ID:201][ROOM_PATH:23,19,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
    CASE 2 TO 4
        _DIC_HO_MAPDATA:MAP_ID:15 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:16 '= "[ROOM_ID:201][ROOM_PATH:17,21,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:17 '= "[ROOM_ID:201][ROOM_PATH:12,16,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:18 '= "[ROOM_ID:201][ROOM_PATH:19,23,][ROOM_NAME:洞穴][EVENT_LIST:21,]"
        _DIC_HO_MAPDATA:MAP_ID:19 '= "[ROOM_ID:201][ROOM_PATH:18,][ROOM_NAME:岩壁][EVENT_LIST:22,]"
        _DIC_HO_MAPDATA:MAP_ID:20 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:21 '= "[ROOM_ID:201][ROOM_PATH:16,22,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:22 '= "[ROOM_ID:200][ROOM_PATH:21,23,][ROOM_NAME:森の入り口]"
        _DIC_HO_MAPDATA:MAP_ID:23 '= "[ROOM_ID:201][ROOM_PATH:18,22,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:24 '= "[ROOM_ID:0]"
    CASE 5 TO 7
        _DIC_HO_MAPDATA:MAP_ID:15 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:16 '= "[ROOM_ID:201][ROOM_PATH:21,][ROOM_NAME:花畑][EVENT_LIST:20,]"
        _DIC_HO_MAPDATA:MAP_ID:17 '= "[ROOM_ID:201][ROOM_PATH:12,18,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:18 '= "[ROOM_ID:201][ROOM_PATH:17,19,][ROOM_NAME:洞穴][EVENT_LIST:21,]"
        _DIC_HO_MAPDATA:MAP_ID:19 '= "[ROOM_ID:201][ROOM_PATH:18,24,][ROOM_NAME:岩壁][EVENT_LIST:22,]"
        _DIC_HO_MAPDATA:MAP_ID:20 '= "[ROOM_ID:0]"
        _DIC_HO_MAPDATA:MAP_ID:21 '= "[ROOM_ID:201][ROOM_PATH:16,22,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:22 '= "[ROOM_ID:200][ROOM_PATH:21,23,][ROOM_NAME:森の入り口]"
        _DIC_HO_MAPDATA:MAP_ID:23 '= "[ROOM_ID:201][ROOM_PATH:22,24,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:24 '= "[ROOM_ID:201][ROOM_PATH:19,23,][ROOM_NAME:獣道][EVENT_LIST:4,12,]"
    CASEELSE
        _DIC_HO_MAPDATA:MAP_ID:15 '= "[ROOM_ID:201][ROOM_PATH:16,20,][ROOM_NAME:下り坂][EVENT_LIST:3,4,]"
        _DIC_HO_MAPDATA:MAP_ID:16 '= "[ROOM_ID:201][ROOM_PATH:15,17,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:17 '= "[ROOM_ID:201][ROOM_PATH:12,16,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:18 '= "[ROOM_ID:201][ROOM_PATH:19,23,][ROOM_NAME:洞穴][EVENT_LIST:21,]"
        _DIC_HO_MAPDATA:MAP_ID:19 '= "[ROOM_ID:201][ROOM_PATH:18,24,][ROOM_NAME:岩壁][EVENT_LIST:22,]"
        _DIC_HO_MAPDATA:MAP_ID:20 '= "[ROOM_ID:201][ROOM_PATH:15,21,][ROOM_NAME:上り坂][EVENT_LIST:3,4,]"
        _DIC_HO_MAPDATA:MAP_ID:21 '= "[ROOM_ID:201][ROOM_PATH:20,22,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:22 '= "[ROOM_ID:200][ROOM_PATH:21,23,][ROOM_NAME:森の入り口]"
        _DIC_HO_MAPDATA:MAP_ID:23 '= "[ROOM_ID:201][ROOM_PATH:18,22,24,][ROOM_NAME:隘路][EVENT_LIST:3,4,12,]"
        _DIC_HO_MAPDATA:MAP_ID:24 '= "[ROOM_ID:201][ROOM_PATH:19,23,][ROOM_NAME:花畑][EVENT_LIST:20,]"
ENDSELECT

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;会話・文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_40
RESULTS '= "MAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_40

CALL QST_QUEST_40_ENEMY_DATA

;コモンイベントをランダムに配置する
;採集
CALL QST_SET_RANDOM_EVENT(3, (RAND:2)+ 1)
CALL QST_SET_RANDOM_EVENT(5, (RAND:3)+ 1)
;遭遇戦闘
CALL QST_SET_RANDOM_EVENT(4, (RAND:3)+ 2)
CALL QST_SET_RANDOM_EVENT(6, (RAND:4)+ 2)
;回復＆Ｈ回避しながらＴＳを進める用イベント
CALL QST_SET_RANDOM_EVENT(12, 3)

FLAG:クエスト_中断を許可 = 1

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_40_ABORT
QST_COOLTIME:QST_有効クエストID += 5
;精算処理へ
CALL QST_QUEST_40_EVENT_25

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストの場合当この関数のみを実行します
;-------------------------------------------------
@QST_INTRO_QUEST_40
CALL PRINT_TITLE("樹精の森", "QUEST")
WAIT
PRINTFORML 辺境の森の奥へと進み続けていた%QST_LEADER_NAME()%は、辺りの雰囲気が変わったことに気付いた
PRINTFORML 見慣れない広葉樹ばかりになり、風もないのに囁くように枝葉が揺れ続けている
PRINTFORML 常に周囲からの視線も感じる……
PRINTL
PRINTFORML どうやら樹精の森に辿り着いたようだ
PRINTW
PRINTFORML 商人から依頼された希少な錬金素材は、主にドライアドと呼ばれる樹木の精霊から採れるものだ
PRINTFORML ドライアドの居る森は豊かに繁茂するため、必然的に様々な魔物が棲みつく
PRINTFORML 慎重に探索するべきだろう……
PRINTW
PRINTFORML 素材さえ手に入れれば、森の最奥にまで分け入ってドライアドを狩る必要もない
PRINTFORML 限界を感じたら入り口に戻り、%QST_LEADER_NAME()%の判断で帰路についても良い
PRINTW
CALL COLOR_PRINTL(@"希少な錬金素材を入手しても、無事に森を出なければ報酬に換えることはできないのだから", カラー_注意)
PRINTW
CALL FL_PLAYBGM("Forest_1.ogg")

;-------------------------------------------------
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0 ボス戦闘
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;-------------------------------------------------
;実行するイベントの実行内容を記述
;イベント用の関数はTRYCALLFORMで呼ばれるため
;@QST_QUEST_{QUEST_ID}_EVENT_{EVENT_ID}の書式で記述する
;-------------------------------------------------
@QST_QUEST_40_EVENT_0
#DIM CONST EVENT_ID = 0
#DIM CHARA_ID
#DIM HAS_SEDUCTION
#DIM TSF_COUNT
#DIM REWARD_POINT

HAS_SEDUCTION = 0
TSF_COUNT = 0

;テキスト分岐用に誘蛾の芳香持ち、元男性の女性が居るかを判定
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
	CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
	CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
	IF RESULT >= 2
		HAS_SEDUCTION = 1
	ENDIF
	IF TS_IS_F(CHARA_ID) >= 1
		TSF_COUNT = 1
	ENDIF
NEXT

CALL FL_STOPBGM()

CALL COLOR_PRINTL(@"「来たか……フフフ、よくぞここまで来られたものだ」", カラー_パ緑)
PRINTW
PRINTFORML 暗い森の深奥で何かが蠢いている……
PRINTFORML 樹々の葉の隙間から僅かに差し込む陽の光に、ゆっくりと３人の人影が歩み出てきた
PRINTW
CALL COLOR_PRINTL(@"「ようこそ、我が閨へ」", カラー_パ緑)
CALL COLOR_PRINTL(@"「眷属達をこうも打ち倒せたのならば、ワタクシの花となれる強さはあるのだろうな」", カラー_パ緑)
PRINTW
CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「キャハハ♪　でも、ワタシ達よりはきっと弱いの♪」@/C@", 1)
CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「だから、あの、よわよわなヒトは、早く降参したほうがいいの……！」@/C@", 1)
PRINTW
PRINTFORML 魔物が棲む森には不似合いな、少女が２人と、女がひとり
PRINTFORML 彼女達が人外の存在であることは、３人に共通した緑色の肌を見れば明らかだ
PRINTW

IF HAS_SEDUCTION >= 1
    CALL COLOR_PRINTL(@"「フフ、この匂い……芳しき牝の香❤」", カラー_パ緑)
    CALL COLOR_PRINTL(@"「ワタクシの花となるには資格も充分。ならば語る時間も惜しいというもの……❤」", カラー_パ緑)
ELSE
    CALL COLOR_PRINTL(@"「ふむ……しかし、芳しき牝の香を感じんな。これではワタクシの花には足りないが……」", カラー_パ緑)
    CALL COLOR_PRINTL(@"「ここまで来たのだ、相手はしてやる。香らぬ蕾を手折ってみるのも、暇潰しになるだろう」", カラー_パ緑)
ENDIF
PRINTW
CALL FL_PLAYBGM("Boss_8.ogg")
PRINTFORMW 中心に立つ女が身振りで合図すると、周囲から太い蔓達が躍り出てきた！
PRINTFORML いままでの道のりで見掛けた枝蔓の魔物達よりも手強そうだ……
PRINTFORML 少女二人も女を守るように、恭しく前へ出て来る
PRINTW
CALL COLOR_PRINTL(@"「ワタクシこそが、今代の樹精が女王ドライアド！」", カラー_パ緑)
CALL COLOR_PRINTL(@"「この身を昂ぶらせるに足るほどには保ってみせよ！」", カラー_パ緑)
PRINTW

CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精女王ドライアド", "樹精女王ドライアド")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精の姫花", "陽気な姫花")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精の姫花", "内気な姫花")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_女王の護り蔓", "女王の護り蔓１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_女王の護り蔓", "女王の護り蔓２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_女王の護り蔓", "女王の護り蔓３")
;戦闘イベント呼出
CALL CLOSE_COMBAT()

IF RESULT == 1
    CALL FL_STOPBGM()
    ;戦闘で増減したステータスをキャラに反映
    CALL CC_APPLY_CHARA_HP()
    ;戦利品獲得処理
    CALL CC_GET_LOOT_FROM("DEFENDER")
    CALL COLOR_PRINTL(@"「ワ、ワタクシが……？」", カラー_パ緑)
    CALL COLOR_PRINTW(@"「ワタクシの身が朽ちてゆく！　枝が、枯れてゆく……なぜ……なぜだ……！」", カラー_パ緑)
    CALL COLOR_PRINTL(@"「なぜヒトがワタクシに勝てる！？　樹精の女王たる……ワ……ワタクシに……」", カラー_パ緑)
    PRINTW
    PRINTFORML 森の樹々が、枝が、葉が、%QST_LEADER_NAME()%を恐れるように退いてゆく
    PRINTFORML 薄暗かった森の一画が、ぽっかりと眩しいほどの陽光に照らされた
    PRINTW
    CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「ああ……姉様……姉様……やだ……やだ……消えたくないの……」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「ヒ、ヒト……なんて……ワタシ達の……ペットのはず……なのに……」@/C@", 1)
    PRINTW
    CALL COLOR_PRINTL(@"「こんな……ことで……終わり……たくな……」", カラー_パ緑)
    CALL COLOR_PRINTW(@"「やっと女王……に……なれ……た……の……に……」", カラー_パ緑)
    CALL COLOR_PRINTL(@"「…………………………」", カラー_パ緑)
    PRINTW
    PRINTFORML ドライアドが地面に倒れ伏すと、枯れ果てて崩れた身体はまるで土へ吸収されるように消えていった……
    PRINTFORML ２人の姫花も後を追ったようだ
    PRINTW
    CALL FL_PLAYSE("Fanfare_0.ogg")
    CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%は樹精の女王を打ち破った！")
    PRINTL
    PRINTFORMW ……………………
    PRINTFORMW ………………
    PRINTFORML …………
    PRINTW
    CALL COLOR_PRINTL(@"「なーんだ、やられちゃったんだ、あいつ」", カラー_女双)
    CALL COLOR_PRINTL(@"「あのヒト達が強すぎるんだもの。あいつらじゃ無理だよ」", カラー_女)
    CALL COLOR_PRINTL(@"「ねえ、次は誰が女王やるの？」", カラー_男の娘)
    PRINTW
    CALL COLOR_PRINTL(@"「ワタシ達でやってみる？」", カラー_女双)
    CALL COLOR_PRINTL(@"「お姫様になれるのなら、いいよ♪」", カラー_女)
    CALL COLOR_PRINTL(@"「ワタシ達ならあのヒトにだって負けないよね！」", カラー_男の娘)
    PRINTW
    CALL FL_PLAYBGM("Forest_1.ogg")
    CALL COLOR_PRINTW(@"「ならば、ワタシ達はワタクシに❤」", カラー_パ緑)
    CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「ワタシ達はワタシに♪」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{カラー_黄緑}@「ワタシ達は……ワ、ワタシに……」@/C@", 1)
    PRINTW
    PRINTFORML 例え何度女王が消え去っても、樹精の森が消えることはないだろう……
    PRINTW
    ;ここから本クエスト特有の精算イベント用報酬入手処理
    REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
    CALL COLOR_PRINTL(@"樹精の種子を手に入れた！(報酬評価:{REWARD_POINT}+100)", カラー_注意)
    PRINTW
    REWARD_POINT += 100
    CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    ;精算イベント用報酬入手処理ここまで
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1
    TFLAG:クエスト_完了フラグ = 1
    QST_COOLTIME:QST_有効クエストID += 5
    CALL QST_QUEST_40_EVENT_25
    CALL TRAIN_EXIT()
ELSE
    CALL CC_APPLY_CHARA_HP()
    CALL QST_QUEST_40_EVENT_45(HAS_SEDUCTION, TSF_COUNT)
ENDIF

;-------------------------------------------------
;EV1 川岸
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_1
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_1
#DIM CONST EVENT_ID = 1

PRINTFORML 樹々が途切れて視界が広がると、%QST_LEADER_NAME()%の前に川が現れた
PRINTFORML 幅は狭くないものの、浅いところを選べば問題なく歩いて渡れそうだ
PRINTW
PRINTFORML ただ、川を境にして向こう側はさらに樹々の密度が増しているのか、薄暗く、異様な雰囲気だ
PRINTFORMW %QST_LEADER_NAME()%の直感が警戒心を新たにさせる
CALL COLOR_PRINTL(@"この先に潜む魔物はこれまでよりも強力かもしれない……", カラー_注意)
PRINTW

;-------------------------------------------------
;EV2 川岸
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_2
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "川原"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_2
#DIM CONST EVENT_ID = 2
#DIM TARGET_ID

PRINTFORML %ANAME(QST_RANDCHARA())%はこの場所ならば周囲の見通しが良いことに気付いた
PRINTFORML 樹々や草藪に囲まれた獣道とは違って、おかしな気配も感じられない
PRINTFORML しばらくの間ならば魔物に見つからず、安全に過ごせそうだ
PRINTW
PRINTFORML 少し休憩するにはちょうど良さそうに見える
PRINTFORMW 消耗しているならば、小休止するのも良いだろう
PRINTL
PRINTFORML 機会は一度きりだが、身体を休めることができそうだ……
PRINTL
CALL ASK_YN("休憩する", "やめておく")
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%は川原の岩に腰掛け、小休憩を取った
    ;パーティ全員の体力を回復
    FOR LOCAL, 0, QST_PARTYNUM
        TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
        CALL QST_CHANGE_HP(TARGET_ID, 1500)
    NEXT
    WAIT
    CALL QST_DEACTIVATE_THIS_EVENT()
ELSE
ENDIF

;-------------------------------------------------
;EV3 南側アイテムイベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "紅蓮の広葉樹"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_3
#DIM CONST EVENT_ID = 3
#DIM DYNAMIC MOB_ID
#DIM CHARA_ID
#DIM REWARD_POINT

PRINTFORMW 周囲の様子を観察していた%ANAME(QST_RANDCHARA())%が、一際目を惹く色鮮やかな木の葉が繁っている樹を見つけた
PRINTFORML 樹精の森では特に精気や栄養を蓄えた葉は美しい暖色に変じるという
PRINTFORML 商人が言っていた錬金素材のひとつ、樹精の葉と呼ばれて珍重される物に間違いない
PRINTW
PRINTFORML よく見れば、樹精の葉を守るように周囲の樹々の枝や蔓が重なっているようだ
PRINTFORML 何本かの枝は風もないのに揺れているようにも感じられる……
PRINTW

CALL QST_ROLL_MULTI("器用,観察,", "枝を避けて採集する(器用),怪しい枝を見極める(観察),","100,100,", , , , , 1, "枝を無視して手を伸ばす(確定失敗)")
CHARA_ID = RESULT:1

IF RESULT == 1
    PRINTFORML %ANAME(CHARA_ID)%は不気味に蠢く枝や蔓に絡め取られることなく、目的の物を採集した
    PRINTW
    REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
    CALL COLOR_PRINTL(@"樹精の葉を手に入れた！(報酬評価:{REWARD_POINT}+10)", カラー_注意)
    PRINTW
    REWARD_POINT += 10
    CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    CALL QST_DEACTIVATE_THIS_EVENT()
ELSEIF RESULT == 0
    ;失敗したら南北共通短時間調教イベントへ
    CALL QST_QUEST_40_EVENT_41(CHARA_ID, EVENT_ID)
    CALL QST_DEACTIVATE_THIS_EVENT()
ENDIF

;-------------------------------------------------
;EV4 南側ランダム戦闘
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_4
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_4
#DIM CONST EVENT_ID = 4
#DIM CHARA_ID
#DIM EV_PATTERN
#DIM REWARD_POINT

EV_PATTERN = 0

CALL HO_GET_ROOM_NAME(HO_MASTER現在地)
LOCALS = %RESULTS%
SELECTCASE RAND:3
    CASE 0
        CALL PRINT_NOTICE(@"敵襲だ！",,1)
        PRINTFORML %QST_LEADER_NAME()%は武器を握りしめ、敵に対峙した
        PRINTW
    CASE 1
        CALL PRINT_NOTICE(@"敵意を持った何かに行く手を塞がれた！",,1)
        PRINTFORML 突破するには眼前の驚異を打ち払わねばならない
        PRINTW
    CASEELSE
        CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%が%LOCALS%に足を踏み入れるなり、草藪から何かが飛び出してきた！")
        PRINTL
ENDSELECT

CALL CC_INIT()
CALL CC_ADDPARTY()
SELECTCASE RAND:100
    CASE 0 TO 24
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_若樹", "立ちはだかる若樹")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_細枝触手", "しなる細枝触手１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_細枝触手", "しなる細枝触手２")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_細枝触手", "しなる細枝触手３")
        ;クエスト用報酬が拾える場合はEV_PATTERNを１に
        EV_PATTERN = 1
    CASE 25 TO 49
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_若樹", "立ちはだかる若樹１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_若樹", "立ちはだかる若樹２")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "野うさぎの群れ", "野うさぎの群れ")
        EV_PATTERN = 1
    CASE 50 TO 64
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ゴブリン", "はぐれゴブリン")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワーグ", "ワーグ")
    CASE 65 TO 79
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_若樹", "立ちはだかる若樹")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_細枝触手", "しなる細枝触手１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_細枝触手", "しなる細枝触手２")
        EV_PATTERN = 1
    CASE 80 TO 94
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ２")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ３")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ４")
    CASEELSE
        ;５％で変な戦闘発生
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_南側フェアリー", "水色エプロンのフェアリー")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_南側フェアリー", "黄色エプロンのフェアリー")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_南側フェアリー", "橙色エプロンのフェアリー")
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「ヒトだー！　いけー、やっつけろー！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「ボク達と遊んで！　たたかいごっこしよう！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「しゅばばばば！　さささっ！　ずぎゅーん！」@/C@", 1)
        PRINTW
        EV_PATTERN = 2
ENDSELECT
CALL CLOSE_COMBAT()

IF RESULT == 1
    CALL CC_APPLY_CHARA_HP()
    CALL CC_GET_LOOT_FROM("DEFENDER")
    IF EV_PATTERN == 2
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「うぎゅー……負けちゃったぁ……」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「本気でやるなんてー！　ごっこなのに！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「てったい！　てったいー！」@/C@", 1)
        PRINTW
    ELSEIF EV_PATTERN == 1
        REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
        CALL COLOR_PRINTL(@"樹精の若枝を手に入れた！(報酬評価:{REWARD_POINT}+4)", カラー_注意)
        PRINTW
        REWARD_POINT += 4
        CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    ENDIF
ELSE
    CALL CC_APPLY_CHARA_HP()
    IF EV_PATTERN == 2
        ;フェアリーパターンでわざと負けると勝つより良いことがあります
        ;負け癖スキルのちょっと変な使い方として実験的に
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「いえーい！　ボク達の強さ、思い知ったかー！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「遊んでくれてありがとう！　もしかして、怪我させちゃった？」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「ごめんね、治してあげるから。あと、これもあげる！」@/C@", 1)
        PRINTW
        PRINTFORML フェアリー達から放たれた暖かな光が、%QST_LEADER_NAME()%を包み込む……
        PRINTW
        FOR LOCAL, 0, QST_PARTYNUM
            CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
            CALL QST_CHANGE_HP(CHARA_ID, 500)
        NEXT
        WAIT
        REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
        CALL COLOR_PRINTL(@"樹精の髪飾りを手に入れた！(報酬評価:{REWARD_POINT}+8)", カラー_注意)
        PRINTW
        REWARD_POINT += 8
        CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    ELSE
        CALL QST_QUEST_40_DEFEAT_EVENT()
    ENDIF
ENDIF

;-------------------------------------------------
;EV5 北側アイテムイベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_5
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "人面洞の広葉樹"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_5
#DIM CONST EVENT_ID = 5
#DIM DYNAMIC MOB_ID
#DIM CHARA_ID
#DIM REWARD_POINT

PRINTFORMW 周囲の様子を観察していた%ANAME(QST_RANDCHARA())%が、奇妙にねじくれた木の実を見つけた
PRINTFORML 樹精の森では稀に、精気が豊富で様々な薬の元になる木の実が見つかるという
PRINTFORML 商人が言っていた錬金素材のひとつ、樹精の踊り果実と呼ばれて珍重される物に間違いない
PRINTW
PRINTFORML よく見れば、木の実を守るように周囲の樹々の枝や蔓が重なっているようだ
PRINTFORML 何本かの枝は風もないのに揺れているようにも感じられる……
PRINTW

CALL QST_ROLL_MULTI("器用,観察,", "枝を避けて採集する(器用),怪しい枝を見極める(観察),","300,300,", , , , , 1, "枝を無視して手を伸ばす(確定失敗)")
CHARA_ID = RESULT:1

IF RESULT == 1
    PRINTFORML %ANAME(CHARA_ID)%は不気味に蠢く枝や蔓に絡め取られることなく、目的の物を採集した
    PRINTW
    REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
    CALL COLOR_PRINTL(@"樹精の踊り果実を手に入れた！(報酬評価:{REWARD_POINT}+30)", カラー_注意)
    PRINTW
    REWARD_POINT += 30
    CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    CALL QST_DEACTIVATE_THIS_EVENT()
ELSEIF RESULT == 0
    ;失敗したら南北共通短時間調教イベントへ
    CALL QST_QUEST_40_EVENT_41(CHARA_ID, EVENT_ID)
    CALL QST_DEACTIVATE_THIS_EVENT()
ENDIF

;-------------------------------------------------
;EV6 北側ランダム戦闘
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_6
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_6
#DIM CONST EVENT_ID = 6
#DIM CHARA_ID
#DIM EV_PATTERN
#DIM REWARD_POINT

EV_PATTERN = 0

CALL HO_GET_ROOM_NAME(HO_MASTER現在地)
LOCALS = %RESULTS%
SELECTCASE RAND:3
    CASE 0
        CALL PRINT_NOTICE(@"敵襲だ！",,1)
        PRINTFORML %QST_LEADER_NAME()%は武器を握りしめ、敵に対峙した
        PRINTW
    CASE 1
        CALL PRINT_NOTICE(@"敵意を持った何かに行く手を塞がれた！",,1)
        PRINTFORML 突破するには眼前の驚異を打ち払わねばならない
        PRINTW
    CASEELSE
        CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%が%LOCALS%に足を踏み入れるなり、草藪から何かが飛び出してきた！")
        PRINTL
ENDSELECT

CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
SELECTCASE RAND:100
    CASE 0 TO 24
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_大樹", "そびえ立つ大樹")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手２")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手３")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手４")
        EV_PATTERN = 1
    CASE 25 TO 49
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_大樹", "そびえ立つ大樹")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精花", "歌う樹精花")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_縄蔓触手", "狂乱の縄蔓触手２")
        EV_PATTERN = 1
    CASE 50 TO 64
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "猛獣使い", "手練れのはぐれゴブリン")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ムマキル", "訓練されたワーグ")
    CASE 65 TO 79
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精花", "笑う樹精花")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_樹精花", "泣く樹精花")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "グール", "樹精花の操り人形♂")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "グール", "樹精花の操り人形♀")
        EV_PATTERN = 1
    CASE 80 TO 94
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_キノコの王", "キノコの王")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ１")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ２")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ３")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ４")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_ころがりキノコ", "ころがりキノコ５")
    CASEELSE
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_北側フェアリー", "水色エプロンのフェアリー")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_北側フェアリー", "黄色エプロンのフェアリー")
        CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_40_北側フェアリー", "橙色エプロンのフェアリー")
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「ヒトだー！　いけー、やっつけろー！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「ボク達と遊んで！　たたかいごっこしよう！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「しゅばばばば！　さささっ！　ずぎゅーん！」@/C@", 1)
        PRINTW
        EV_PATTERN = 2
ENDSELECT
CALL CLOSE_COMBAT()

IF RESULT == 1
    CALL CC_APPLY_CHARA_HP()
    CALL CC_GET_LOOT_FROM("DEFENDER")
    IF EV_PATTERN == 2
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「うぎゅー……負けちゃったぁ……」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「本気でやるなんてー！　ごっこなのに！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「てったい！　てったいー！」@/C@", 1)
        PRINTW
    ELSEIF EV_PATTERN == 1
        REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
        CALL COLOR_PRINTL(@"樹精の花枝を手に入れた！(報酬評価:{REWARD_POINT}+12)", カラー_注意)
        PRINTW
        REWARD_POINT += 12
        CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    ENDIF
ELSE
    CALL CC_APPLY_CHARA_HP()
    IF EV_PATTERN == 2
        ;ここも負けるが勝ち
        CALLF TAG_PRINT(@"@C:{カラー_シアン}@「いえーい！　ボク達の強さ、思い知ったかー！」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_黄}@「遊んでくれてありがとう！　もしかして、怪我させちゃった？」@/C@", 1)
        CALLF TAG_PRINT(@"@C:{カラー_オレンジ}@「ごめんね、治してあげるから。あと、これもあげる！」@/C@", 1)
        PRINTW
        PRINTFORML フェアリー達から放たれた暖かな光が、%QST_LEADER_NAME()%を包み込む……
        PRINTW
        FOR LOCAL, 0, QST_PARTYNUM
            CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
            CALL QST_CHANGE_HP(CHARA_ID, 1000)
        NEXT
        WAIT
        REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
        CALL COLOR_PRINTL(@"樹精の服を手に入れた！(報酬評価:{REWARD_POINT}+24)", カラー_注意)
        PRINTW
        REWARD_POINT += 24
        CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
    ELSE
        CALL QST_QUEST_40_DEFEAT_EVENT()
    ENDIF
ENDIF

;-------------------------------------------------
;EV7 小屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_7
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "壊れかけの本棚"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_7
#DIM CONST EVENT_ID = 7

;そのうち文章追加するかも……しないかも……
PRINTL
PRINTFORML 樹々の間に、危険な森には不似合いな木製の小屋が建っている
PRINTFORMW 赤い屋根に黄色い壁、カラフルに装飾されたかわいらしい小屋だ
PRINTFORML ただ、今はもう誰も住んでいないのか、屋根も壁も穴が空いて壊れかけている
PRINTW
PRINTFORMW 屋内へ入ればベッドらしき物、他の家具らしき物の残骸が転がっている
PRINTFORML 唯一、本棚だけは比較的状態が良いようだ
PRINTFORML 中を覗いてみると、まだ読めそうな手書きの冊子がいくつか置かれてあった
PRINTW
$BOOK_CHOICE
PRINTFORML どれを読みますか？
PRINTL
CALL ASK_MULTI("樹精の森案内書", "要注意スポットメモ", "日記", "どれも読まない")
PRINTL
IF　RESULT == 0
    SETCOLOR 0xffecc3
    PRINTFORML 樹精の森について
    PRINTL
    PRINTFORML 言うまでもなく、ドライアドの名で知られる樹木の精霊を女王に据えた森だ
    PRINTFORML だが冒険者が言うようにドライアドが棲む森、ではない
    PRINTFORML ドライアドが森なのだ
    PRINTFORML 少なくとも樹精の森のドライアドは、樹精の森そのものである
    PRINTFORML いわば森はドライアドの体内というわけだ
    PRINTFORML だから女王の気分で毎日、樹々の配置が変わる
    PRINTL
    PRINTFORML そんな森の中に棲むのは樹精だけではない
    PRINTFORML 彼女達は意外と寛容な部分もあり、森を荒らしさえしなければゴブリンも受け入れる
    PRINTFORML 私も彼女達に受け入れられ、樹精の森に住まうことを許された
    PRINTL
    PRINTFORML だがひとたび森を荒らそうという意思を持って踏み入れば……
    PRINTFORML ほとんどの冒険者は森の特異な植物を採取しようと分け入るので、彼女達は例外なく敵視する
    PRINTFORML また、彼女達が牝香と呼ぶ、才能あるいは特徴のある人物はさらに危険だ
    PRINTFORML 弄び、嬲り、淫らな呪いの実験台にする
    PRINTFORML そして最終的には森へ取り込み、樹精の仲間にしようとするのだ
    PRINTL
    PRINTFORML では、森で採れる特徴的な植物素材についてだが、まず
    PRINTFORMW
    RESETCOLOR
    PRINTFORML ……この先はページが破れている
    PRINTW
    GOTO BOOK_CHOICE
ELSEIF RESULT == 1
    SETCOLOR 0xffecc3
    PRINTFORML 要注意スポット
    PRINTFORML 冒険者に売れそうな情報をメモ、情報は金になる
    PRINTL
    PRINTL
    PRINTFORML 女王の寝室：
    PRINTFORML 変態女の寝床、気取って閨とか言ってる
    PRINTFORML 女王は好色だから森が冒険者を捕まえたら性的に弄ぶ
    PRINTFORML ただし処女性に拘る変態だから、キスとかも含めて、徹底的に「初めて」を奪う可能性のあることはしない
    PRINTFORML だけどこの寝室で女王に捕まったら別
    PRINTFORML もしも捕まった中にひとりでも牝香の才能がある人が居れば、興奮した女王に「初めて」を奪われてしまう
    PRINTFORML その場合は牝香の才能がない人も巻き添えで、変態ちんちん女に犯されるかも
    PRINTL
    PRINTL
    PRINTFORML フェアリーの舞台：
    PRINTFORML たまに森にできてる、庭園みたいに綺麗なところ
    PRINTFORML 森に居候しているフェアリーの中でもタチ悪めな子達がよく踊ってる
    PRINTFORML 牝香とかいう才能があると踊りに誘われて、胸やお尻を大きくされるらしい
    PRINTL
    PRINTL
    PRINTFORML フェアリーの台所：
    PRINTFORML 沼地になっているところがたまーに森の機嫌で池になってることがある
    PRINTFORML ここでよくお菓子作りをしているフェアリーくん達は良い子なんだけど……比較的
    PRINTFORML 牝香のある冒険者には、子どもにする悪戯をしてるみたいだ
    PRINTL
    PRINTFORML あと、ときどきこの子達は戦いごっことか言って冒険者に絡んでる
    PRINTFORML 悪い子達じゃないから、もし同じことをされたら負けたフリでもしてあげてほしい
    PRINTL
    PRINTL
    PRINTFORML リンゴみたいな実がひとつだけ生ってる樹：
    PRINTFORML 樹精達の罠……なんだけど牝香の才能がなければただのおいしい果物だ
    PRINTFORML ただし、黄色いやつは食べないほうがいいかも
    PRINTFORML お菓子好きのフェアリーくん達がやる悪戯と似た呪いが掛けられてる
    PRINTFORML しかも身体が元から弱い人は、たまに呪いをもろに浴びちゃうみたい
    PRINTW
    RESETCOLOR
    PRINTFORML 次のページへ
    SETCOLOR 0xffecc3
    PRINTW
    PRINTFORML 洞窟：
    PRINTFORML スライムみたいな奴が棲み着いてる
    PRINTFORML ……ように見えるけど、コイツも下等な樹精の一種
    PRINTFORML 何とかできる自信がないなら襲われても下手に暴れないほうがいいかも
    PRINTL
    PRINTL
    PRINTFORML 洞窟の外の崖：
    PRINTFORML フェアリー達が宝石屋さんごっこに使ってる崖
    PRINTFORML 冒険者が盗むと怪盗が来たとか言って喜ぶから、遠慮なく盗んであげてほしい
    PRINTFORML 本当に宝石置いてるところは見たことないけど
    PRINTL
    PRINTL
    PRINTFORML 森の入り口に近い花畑：
    PRINTFORML 多分、牝香とかいう才能がある冒険者にとっては最大の要注意スポット
    PRINTFORML 女王は最近、獲物をより可愛らしくする呪いに傾倒している
    PRINTFORML 端的には男性を女っぽくして、最終的には本当に女にしたいらしい
    PRINTFORML 可愛い男の子を犯したい変態性癖が高じたんだろう
    PRINTL
    PRINTFORML 実際に男性を女性にしたところはあたしもまだ見たことない
    PRINTFORML でも、女性化の呪いの研究にはあたしも協力させられた。森に住む対価として
    PRINTFORML だから言えるんだけど、おそらく条件さえ揃えば呪いが成功する可能性はある
    PRINTFORML ↑これやっぱり書かなくていいな。もし冒険者にバレたらあたしが恨まれるかも
    PRINTW
    RESETCOLOR
    GOTO BOOK_CHOICE
ELSEIF RESULT == 2
    SETCOLOR 0xffecc3
    PRINTFORML 日記
    PRINTL
    PRINTFORML フェアリーくん達へ
    PRINTFORML 家を造ってくれたことには感謝してる
    PRINTFORML 変態樹精達があたしに手出ししないように、毎日守ってくれてるのも
    PRINTFORML でも、もしまたこの日記を勝手に読んだら……
    PRINTFORML もうマフィンは作ってあげないし、次はお尻ぺんぺんではすまさない
    PRINTFORML （怒っている顔の絵）
    PRINTW
    RESETCOLOR
    PRINTFORML ……表紙だけで中身は抜き取られていた
    PRINTW
    GOTO BOOK_CHOICE
ENDIF

;-------------------------------------------------
;EV9 空き地
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_9
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_9
#DIM CONST EVENT_ID = 9

;男性化の需要があればここから男性化クエストの情報を得て……とか追加を作るかも……作らないかも……作らないかも……
PRINTL
PRINTFORML 森の奥を進んでいた%QST_LEADER_NAME()%は、樹々の途切れた広い空間に出食わした
PRINTFORML 下草も生えていない空き地には、ひとりの人影だけが立っていた
PRINTW
CALL COLOR_PRINTW(@"「うわ、冒険者！？　こんなところまで……！」", カラー_ダークオレンジ)
PRINTW
PRINTFORMW 人影――緑肌の若い女性は、%QST_LEADER_NAME()%に気付くと慌てて逃げていってしまった
PRINTFORML 肌の色からして森に棲む人外の存在に違いない
PRINTFORML %QST_LEADER_NAME()%は辺りに警戒しながら、何もない空き地を後にした
PRINTW

;-------------------------------------------------
;EV10 吊り橋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_10
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_10
#DIM CONST EVENT_ID = 10

PRINTFORMW 森を分かつ断崖の間に、奇妙な作りをした橋が掛けられている
PRINTFORML ねじれて絡み合った樹の枝が綱のように伸びて、太い蔦がそれを支える吊り橋だ
PRINTFORML 複雑に編み込まれた枝が床の役目を果たしているおかげで、歩いて渡ることができそうだ
PRINTW
PRINTFORML 人の手によって作られたものではないだろう
PRINTFORML まるでこの先へと誘うように、愉快げに吊り橋は風に揺れている……
PRINTW

;-------------------------------------------------
;EV21 洞穴
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_21
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_40_EVENT_21
#DIM CONST EVENT_ID = 21
#DIM TARGET_ID
#DIM REWARD_POINT

TARGET_ID = QST_RANDCHARA()

PRINTFORMW 岩壁を穿つ洞穴の中を一行が進んでいると、天井から落ちてきた水滴が大きな音を立てた
PRINTFORML 暗く湿った岩穴の中、水滴の音は続けて何度も響き続けている
PRINTFORML 気のせいか、水滴が落ちてくる頻度も音の大きさも、徐々に増しているように感じられる……
PRINTW
PRINTFORMW 洞穴の中央にまで入ったところで、%ANAME(TARGET_ID)%の頭上からそれは襲い掛かった
PRINTFORML 天井から滲み出て糸を引く、巨大な水滴にも見えるそれは意思を持つ粘液の魔物、スライムだ！
PRINTW
CALL PRINT_TITLE("樹液のスライム", "エロイベント")
WAIT
PRINTFORMW %ANAME(TARGET_ID)%は反射的に振り払おうとするが、粘液状の相手には手応えが弱い
PRINTFORML この掴みどころのない魔物を撃退するには……
PRINTW
CALL COLOR_PRINTW(@"―――― 大きなスライムが%ANAME(TARGET_ID)%を頭から包み込もうとしている！", カラー_注意)
PRINTL
CALL QST_ROLL_MULTI("知識,戦闘,", "適切に対処する(知識),がむしゃらに撃退する(戦闘),","100,400,", , ,"なすがまま", TARGET_ID)
PRINTL
IF RESULT == 1 && RESULT:2 == 0
    PRINTFORML %ANAME(TARGET_ID)%はとっさに地面の泥をすくい取り、スライムの体内にねじ込んだ
    PRINTFORML 予期せぬ異物が溶け込むのを嫌がったスライムの意識が%ANAME(TARGET_ID)%から逸れる
    PRINTFORML その一瞬の隙をついて、%ANAME(TARGET_ID)%はスライムを振り払うことに成功した
    PRINTW
    PRINTFORML スライムはそのまま地面の亀裂に染み込んで逃げてしまった……
    PRINTW
ELSEIF RESULT == 1
    PRINTFORML %ANAME(TARGET_ID)%はとにかく無我夢中に暴れて、強引にスライムを引き剥がそうとした
    PRINTFORML 自身がダメージを受けることも構わず、暴れ回る粘液に攻撃を加えていく
    PRINTFORML 偶然か、%ANAME(TARGET_ID)%の経験の賜物か、スライムのコアにトドメを刺すと、あとには静かな粘液塊が残った……
    PRINTW
    CALL QST_CHANGE_HP(TARGET_ID, -100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_乳首, 100)
    REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
    CALL COLOR_PRINTL(@"樹精の樹液を手に入れた！(報酬評価:{REWARD_POINT}+10)", カラー_注意)
    PRINTW
    REWARD_POINT += 10
    CALL QST_SET_FLAGMAP("EVENT_25_報酬評価", TOSTR(REWARD_POINT))
ELSEIF RESULT == -1
    PRINTFORML 諦めか、好奇心か、それとも何か考えがあってのことなのか……
    PRINTFORML とにかく%ANAME(TARGET_ID)%は抵抗することなく、スライムに身を任せることにした
    PRINTW
    PRINTFORML 無防備な%ANAME(TARGET_ID)%を頭から包み込んだスライムが衣服を溶かしていく
    PRINTFORML 粘度の高いひんやりとした感触が、肌を舐めるように滑り落ちていった
    PRINTFORML ……どういうわけか、それ以上の危害を加えるつもりはないようだ
    PRINTW
    PRINTFORML ただ通りすがっただけだとでもいうように、やがてスライムは地面の亀裂へ消えていった
    PRINTFORML 粘液を浴びた%ANAME(TARGET_ID)%の身体からは、不思議なことに疲労感が失せている
    PRINTFORML 同時に妙に火照ったように熱くなり、露わになった肌には透明な汗の雫が浮いていた……
    PRINTW
    CALL QST_CHANGE_HP(TARGET_ID, 300)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_乳首, 100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_乳首, 100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_秘部, 100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_秘部, 100)
    CALL COLOR_PRINTW(@"―――― %ANAME(TARGET_ID)%は魔物の体液毒により発情してしまった……！", カラー_ピンク)
    TCVAR:TARGET_ID:媚薬 += 1
ELSE
    PRINTFORML %ANAME(TARGET_ID)%はスライムの撃退を試みるも、何をやっても有効打とはならなかった
    PRINTFORML 無駄な抵抗を続ける獲物の身体を滑り落ちていった粘液は、やがて地面の亀裂へと消えていった……
    PRINTW
    CALL QST_CHANGE_HP(TARGET_ID, -100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_乳首, 100)
    CALL QST_ARMOR_BREAK(TARGET_ID, 被覆部位_秘部, 100)
ENDIF

;-------------------------------------------------
;EV22 岩壁
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_22
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "妖精の飾り棚"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_22
#DIM CONST EVENT_ID = 22
#DIM DYNAMIC CHARA_ID

PRINTFORML 進路に立ちはだかる岩壁をふと見上げた%ANAME(QST_RANDCHARA())%が立ち止まった
PRINTW
PRINTFORML なんの変哲もない灰色の岩壁に、何箇所かキラキラと光を反射する部分が見える
PRINTFORML どうやら岩壁の出っ張りに鉱物らしき塊が並べられているようだ
PRINTFORML 動物やただの魔物の仕業ではなく、妖精の遊びの一種なのではないだろうか
PRINTW
PRINTFORML 中には有用な鉱物も混ざっているかもしれない
PRINTFORML 高い場所にあるため、岩壁を登らなければならないが……
PRINTW
CALL PRINT_TITLE("宝石屋さんごっこと怪盗ごっこ", "イベント")
WAIT
CALL ASK_YN("回収を試みる（身体判定）", "放っておく")
IF !RESULT
    CALL QST_PARTY_ROLL("身体", 100, 1, 3)
    CHARA_ID = RESULT:1
    IF RESULT == 1
        PRINTFORML %ANAME(CHARA_ID)%は岩壁の凹凸に手足をかけて、慎重に登り始めた
        PRINTFORML デコボコしているおかげで思ったよりも楽に登っていけそうだ
        PRINTW
        PRINTFORML 塊のほとんどは綺麗なだけの石ころだったが、使えそうな金属鉱物も混ざっている
        PRINTFORML %ANAME(CHARA_ID)%は鉱物を下に落としてから、油断することなく降りていった
        PRINTW
        SELECTCASE RAND:100
            CASE 0 TO 39
                CALL CHANGE_ITEM_STOCK(ITEM_銅鉱石, 3)
                CALL GET_ITEM_NAME(ITEM_銅鉱石)
                CALL COLOR_PRINTW(@"%RESULTS%を３つ手に入れた", カラー_注意)
            CASE 40 TO 84
                CALL CHANGE_ITEM_STOCK(ITEM_鉄鉱石, 3)
                CALL GET_ITEM_NAME(ITEM_鉄鉱石)
                CALL COLOR_PRINTW(@"%RESULTS%を３つ手に入れた", カラー_注意)
            CASE 85 TO 94
                CALL CHANGE_ITEM_STOCK(ITEM_銀鉱石, 2)
                CALL GET_ITEM_NAME(ITEM_銀鉱石)
                CALL COLOR_PRINTW(@"%RESULTS%を２つ手に入れた", カラー_注意)
            CASEELSE
                CALL CHANGE_ITEM_STOCK(ITEM_金鉱石, 1)
                CALL GET_ITEM_NAME(ITEM_金鉱石)
                CALL COLOR_PRINTW(@"%RESULTS%を１つ手に入れた", カラー_注意)
        ENDSELECT
        WAIT
        CALL QST_DEACTIVATE_THIS_EVENT()
    ELSEIF RESULT == 0
        CALL COLOR_PRINTW(@"―――― 順調に登っていた%ANAME(CHARA_ID)%だが、突然、重みに耐えかねた岩壁が崩れて落下してしまった！", カラー_注意)
        CALL QST_CHANGE_HP(CHARA_ID, -300)
        WAIT
        CALL QST_DEACTIVATE_THIS_EVENT()
    ENDIF
ELSE
ENDIF

;-------------------------------------------------
;EV23 樹々の隙間
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_40_EVENT_23
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "草藪の奥"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_40_EVENT_23
#DIM CONST EVENT_ID = 23
#DIM DYNAMIC CHARA_ID

PRINTFORML 行き止まりに繁る草藪の向こうを見て、%ANAME(QST_RANDCHARA())%が立ち止まった
PRINTW
PRINTFORML どういうわけか、動くキノコの魔物達が地面へ何かを埋めているようだ
PRINTFORML 短い手足状の軸部分を器用に動かして、どこか楽しげに土を埋め戻している
PRINTW
PRINTFORML やがて何かを埋め終えたキノコ達は、ころがりながら森の奥へ消えていった
PRINTFORML ……掘り返してみれば、何か有用な物が埋まっているかもしれない
PRINTW
CALL PRINT_TITLE("地に埋まりしは……", "イベント")
WAIT
CALL ASK_YN("キノコ達の宝を探す（観察判定）", "放っておく")
IF !RESULT
    CALL QST_PARTY_ROLL("観察", 100, 1, 3)
    CHARA_ID = RESULT:1
    IF RESULT == 1
        PRINTFORML 埋まっている場所に目星をつけた%ANAME(CHARA_ID)%は、手近な枝を拾って土を払っていく
        PRINTFORML 太い枝を地面に突き刺し、掘り返し続けていると、やがて感触が変わった
        PRINTW
        PRINTFORMW %ANAME(CHARA_ID)%が慎重に広げた穴の中を覗き込むと……
        SELECTCASE RAND:10
            CASE 0 TO 2
                PRINTFORML つるつるとした鳥の卵らしきものが埋まっていた
                PRINTFORML まだ新鮮なもののようだ
                PRINTW
                CALL CHANGE_ITEM_STOCK(ITEM_動物の卵, 3)
                CALL GET_ITEM_NAME(ITEM_動物の卵)
                CALL COLOR_PRINTW(@"%RESULTS%を３つ手に入れた", カラー_注意)
            CASE 3 TO 5
                PRINTFORML 少し土で汚れてはいるが、美しい、ふわふわとした鳥の羽が埋まっていた
                PRINTW
                CALL CHANGE_ITEM_STOCK(ITEM_鳥の羽, 3)
                CALL GET_ITEM_NAME(ITEM_鳥の羽)
                CALL COLOR_PRINTW(@"%RESULTS%を３つ手に入れた", カラー_注意)
            CASE 6 TO 7
                PRINTFORML 束になった小麦が埋まっていた
                PRINTFORML どこに生えていたのだろうか、新鮮なもののようだ
                PRINTW
                CALL CHANGE_ITEM_STOCK(ITEM_小麦, 3)
                CALL GET_ITEM_NAME(ITEM_小麦)
                CALL COLOR_PRINTW(@"%RESULTS%を３つ手に入れた", カラー_注意)
            CASEELSE
                PRINTFORML プン、と強烈な匂いが鼻をついた
                PRINTFORML 穴の中には奇妙な質感をした、香り立つ黒土がたっぷりと溜まっている
                PRINTW
                PRINTFORML ……これは動物の糞だ
                PRINTW
                CALL ASK_YN("有用なので持って帰る", "埋め戻す")
                IF !RESULT
                    CALL CHANGE_ITEM_STOCK(ITEM_動物の糞, 3)
                    CALL GET_ITEM_NAME(ITEM_動物の糞)
                    CALL COLOR_PRINTW(@"……%RESULTS%を３つ手に入れた", カラー_注意)
                ELSE
                    PRINTFORML %ANAME(CHARA_ID)%はそれができるだけ身体や手に触れないよう注意して埋め戻した……
                    PRINTL
                ENDIF
        ENDSELECT
        WAIT
        CALL QST_DEACTIVATE_THIS_EVENT()
    ELSEIF RESULT == 0
        PRINTFORML %ANAME(CHARA_ID)%は草藪の奥を行き来するが、どうもキノコ達が何かを埋めた場所を見失ってしまったようだ
        PRINTFORMW 繁みをかきわけて行ったり来たりしていると、その音がまずかったのか……
        PRINTFORML 戻ってきたキノコ達の一匹がこちらへ走ってきた！
        PRINTW
        PRINTFORML 菌類とは思えない俊敏さで走り、跳び、%ANAME(CHARA_ID)%目掛けて石突きの部分を向けて突っ込んでくる……！
        PRINTW
        CALL COLOR_PRINTW(@"―――― 不意を突かれた%ANAME(CHARA_ID)%は強烈なキックを食らってしまった！", カラー_注意)
        CALL QST_CHANGE_HP(CHARA_ID, -500)
        PRINTW
        PRINTFORML %ANAME(CHARA_ID)%が尻餅をついている間に、他のキノコ達が何かを素早く掘り返して逃げていってしまった……
        PRINTW
        CALL QST_DEACTIVATE_THIS_EVENT()
    ENDIF
ELSE
ENDIF

;-------------------------------------------------
;EV25 報酬精算
;-------------------------------------------------
@QST_QUEST_40_EVENT_25
#DIM CONST EVENT_ID = 25
#DIM DYNAMIC CHARA_ID
#DIM REWARD_POINT
#DIM TOTAL_POINT
#DIM EXCHANGE_NUM
#DIM MONEY_POINT
#DIM HONOR_POINT

REWARD_POINT = TOINT(QST_GET_FLAGMAP("EVENT_25_報酬評価"))
MONEY_POINT = REWARD_POINT * 100
HONOR_POINT = REWARD_POINT / 4

IF REWARD_POINT > 0
    PRINTL
    PRINTFORMW ……………………
    PRINTFORMW ………………
    PRINTFORML …………
    PRINTW
    PRINTFORML 「よう、待ってたぜ！　その顔だとうまくいったみたいじゃないか」
    PRINTW
    PRINTFORML 街へ戻った%QST_LEADER_NAME()%を、今回の依頼者である商人が出迎える
    PRINTFORML %QST_LEADER_NAME()%が樹精の森で見つけた希少な錬金素材を披露すると、すぐに商人は査定をはじめた
    PRINTW
    PRINTFORML 「ふむ、ふむ、ふむ……」
    PRINTFORMW 「よしよし、これなら全部まとめて引き取らせて貰えばこの程度の金額は払えるぞ」
    PRINTFORML 「ああ、それとも……馴染みの錬金術師の集まりがあるんだが」
    PRINTFORML 「あんたさえ良ければ、そいつらに安く譲ってやるって手もあるぜ」
    PRINTW
    PRINTFORML 『交織の蛇』なる錬金術師集団に格安で渡せば、冒険者として良い評価に繋がるということらしい
    PRINTFORMW 代わりに、%QST_LEADER_NAME()%の金銭的な儲けはなくなるだろう
    PRINTFORML ……その場合でも、商人の利益は変わらないようだが
    PRINTW
    PRINTFORML 「それとな、その『交織の蛇』って連中がこの森の研究をしているわけなんだが」
    PRINTFORML 「多く取引してくれるなら、研究成果の一部を分けてやってもいいんだとさ」
    PRINTW
    CALL ASK_MULTI(@"お金に換える({MONEY_POINT}ディナリ)", @"『交織の蛇』に譲る(名誉点+{HONOR_POINT})")
    ;名誉点のディナリに対するレートは隊商護衛の半分にしています
    PRINTL
    IF RESULT == 0
        CALL COLOR_PRINTW(@"――　{MONEY_POINT}ディナリを得た！", カラー_注意)
        MONEY += MONEY_POINT
    ELSE
        CALL COLOR_PRINTW(@"――　名誉点を{HONOR_POINT}点得た！", カラー_注意)
        QST_PRESTIGE_POINT += HONOR_POINT
        CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
        TOTAL_POINT = REWARD_POINT + RESULT
        PRINTL
        IF TOTAL_POINT >= 100
            PRINTFORML 「へへへ、連中に売って……おっと、わけてやる量もこれだけあれば充分だろうな」
            PRINTFORML 「もしあんたに渡すもんを受け取ったら、すぐに届けに行くぜ！」
            PRINTW
            EXCHANGE_NUM = TOTAL_POINT / 100
            TOTAL_POINT -= EXCHANGE_NUM * 100
            CALL CHANGE_ITEM_STOCK(ITEM_樹華結晶, EXCHANGE_NUM)
            CALL GET_ITEM_NAME(ITEM_樹華結晶)
            CALL COLOR_PRINTL(@"後日、返礼として『交織の蛇』から%RESULTS%が{EXCHANGE_NUM}個、届いた！", カラー_注意)
        ELSEIF TOTAL_POINT >= 70
            PRINTFORML 「これだけ納品してやれば、連中もそろそろあんたに礼をくれるんじゃないか？」
            PRINTFORML 「また頼むぜ！」
        ELSEIF TOTAL_POINT >= 30
            PRINTFORML 「いい調子だが、連中はこれでも足りないってうるさいんだ」
            PRINTFORML 「まだまだ頼むぜ！」
        ELSE
            PRINTFORML 「連中を満足させる量には遠いが、あんたが無事に帰ってくるのが一番ってもんだ」
            PRINTFORML 「なんせ生きてりゃまた頼めるからな、頼りにしてるぜ！」
        ENDIF
        PRINTW
        CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, TOTAL_POINT)
    ENDIF
    PRINTL
    PRINTL
ENDIF